

# PIC 12F683 and the ISD 17150 ChipCorder

## Establishing Initial Communications Using PIC Basic Pro

### Overview

Windbond/Nuvoton has a series of inexpensive Demo Boards for the ISD 1700 series ChipCorders. These boards come complete with everything needed to manually record and play back messages. All that is needed is an 8-ohm speaker and a 3 to 5 volt DC power source.

The real potential of the ISD 1700 series ChipCorders is the ability to address various locations in the memory using Serial Peripheral Interface (SPI) commands. These commands would normally be sent via a micro controller – in these examples by a PIC12F683 programmed using the PIC Basic Pro (PBP) Compiler by Micro Engineering Labs and the MicroCode Studio Editor by Mechanique Co.

### Objective

To establish communication from the PIC 12F683 to the ISD-COB 17150 (ISD) using Serial Peripheral Interface (SPI) Commands. This then allows further development of code necessary to control the ISD-COB 17150 Demo Board.

### Process and Experience


In order to confirm communication from the PIC to the ISD, a long passage of music was recorded on the ISD Demo Board using the push buttons and the built-in microphone. Communication would be established if the music could be made to PLAY via commands send by the PIC. Although there are a number of feedback options from the ISD to the PIC, these are not addressed at this time.

Two methods were used to test send data from the PIC to the ISD.


The first method employed ‘bit banging’ and although very tedious, this worked very well and served to better understand the required timing of the data and clock bits. The code used for this is contained in Example 1.

The second method used the SHIFTOUT command in PBP and was very quick and simple. See Examples 2 and 3.

ISD COB 17150  
Demonstration  
Board



Although the bit banging approach initially appears redundant in view of the ease of using SHIFTOUT, it will be very useful when two-way communication is necessary with the ISD1700.



The three output pins on the PIC12F683 that interface with the ISD are GPIO.0 (SS signal), GPIO.1 (SCLK) and GPIO.2 (MOSI). If In-Circuit-Serial-Programming (ICSP) is used with the 12F683, it is necessary to disconnect the Demo Board during programming.

### Example 1 - PIC Basic Pro Code using Bit Banging

```
' Name:   PLAYING the ISD17150 - BB.BAS
' Author: Ian Burn
' Date :  05/02/2009
' Version: lost count!!
' Notes: 209 Words   Special thanks to Jerry Wang of Nuvoton
'         bits are sent with Least Significant Bit first
'
' References:
'         ISD1700 Design Guide.PDF - Timing diagrams KEY TO SUCCESS
'         ISD1700 Data Sheet.PDF
'
' OVERVIEW
'
' Objective is to PLAY a message, that has been pre-recorded on
' the Nuvoton ISD COB 17150 Demo Board, using SPI codes.
' Two commands are sent: POWER-UP and PLAY
' Clock and data bits are sent individually, one at a time. This is very tedious
' but very 'solid'
```

```

'
'-----
'CONFIGURATION
'-----

TRISIO=%001000    'Ports 0,1,2,4 and 5 as OUTPUTS
CMCON0=7          'COMPARATOR OFF
ANSEL=0           'all ports set as digital

@ DEVICE PIC12F683, WDT_ON, MCLR_ON, BOD_OFF, PWRT_ON, PROTECT_OFF

'
'-----
'DEFINE VARIABLES
'-----

'PORTS

SS  VAR GPIO.0    'Pin #7  output
SCLK VAR GPIO.1   'Pin #6  output
MOSI VAR GPIO.2   'Pin #5  output
'GPIO.3 MCLR      'Pin #4  input
LED  VAR GPIO.4   'Pin #3  output
'GPIO.5 not used  'Pin #2  input

'Other Variables

x VAR word        'misc. counter

'
'-----
'PROGRAM
'-----

START:
HIGH LED          'LED ON during the program
GOSUB POWER_UP
pause 500
GOSUB PLAY
LOW LED           'LED OFF

END

'
'-----
'SUBROUTINES
'-----

Lbit:    'low bit for bit banger process
LOW SCLK
PAUSE 10
HIGH SCLK      'shift out 1 at this rising edge
PAUSE 10
return

Hbit:    'high bit for bit banger process
HIGH MOSI
LOW SCLK
HIGH MOSI
PAUSE 10
HIGH SCLK      ' shift out 1 at this rising edge
PAUSE 10
LOW MOSI
return
init:     'initialization at start of bit banded command

```

```

HIGH SS
LOW MOSI
HIGH SCLK
pause 10
LOW SS
Pause 10
return

ENDset: 'clean up at the end of bit banged command
LOW MOSI
HIGH SCLK
HIGH SS
Return

POWER_UP: '10000000 00000000 ($01,$00)
GOSUB init
GOSUB Hbit 'bit 0
for x=1 to 14 'bits 2 to 15 are all low
GOSUB Lbit
next x
GOSUB ENDset
return

PLAY: '00001010 00000000 ($50,$00) LED option on - bit 4
GOSUB init
for x = 1 to 4 'bits 0 to 3 are all low
GOSUB Lbit
next x
GOSUB Hbit 'bit 4 is high set LED on ISD
GOSUB Lbit 'bit 5 is low
GOSUB Hbit 'bit 6 is high
for x = 1 to 9
GOSUB Lbit
next x
GOSUB ENDset
return

```

## Example 2 - PIC Basic Pro Code using SHIFTOUT

```

' Name: PLAYING the ISD17150-SO.BAS
' Author: Ian Burn
' Date : 13/02/2009
' Version: 3
' Notes: 149 Words Special thanks to Charles Leo of Micro Engineering Labs
' MODE 4 in SHIFTOUT command sets clock high and Least Sig Bits sent first
'
'
'References:
' ISD1700 Design Guide.PDF - Timing diagrams KEY TO SUCCESS
' ISD1700 Data Sheet.PDF
'


---


'OVERVIEW
'
' Objective is to PLAY a message, that has been pre-recorded on
' the Nuvoton ISD COB 17150 Demo Board, using SPI codes.
' Two commands are sent: POWER-UP and PLAY
' All timing and data transmission are done via the SHIFTOUT command in PBP

```

```

'
-----
'CONFIGURATION
'
-----

TRISIO=%001000    'Ports 0,1,2,4 and 5 as OUTPUTS
CMCON0=7          'COMPARATOR OFF
ANSEL=0           'all ports set as digital

@ DEVICE PIC12F683, WDT_ON, MCLR_ON, BOD_OFF, PWRT_ON, PROTECT_OFF

'
-----
'DEFINE VARIABLES
'
-----

'PORTS

SS VAR GPIO.0     'Pin #7  output
SCLK VAR GPIO.1   'Pin #6  output
MOSI VAR GPIO.2   'Pin #5  output
'GPIO.3 MCLR      'Pin #4  input
LED VAR GPIO.4    'Pin #3  output
'GPIO.5 not used  'Pin #2  input

'Other Variables

x VAR word        'misc. counter

'
-----
'PROGRAM
'
-----

START:
HIGH LED          'LED ON during the program

HIGH SS
LOW MOSI
HIGH SCLK

'POWER_UP
LOW SS
SHIFTOUT MOSI, SCLK, 4, [$01,$00] 'see page 41 in ISD1700 Design Guide
HIGH SS

PAUSE 10          'Depending on the command that follows this isn't always needed
                  'but is likely good insurance

'PLAY
LOW SS
SHIFTOUT MOSI, SCLK, 4, [$50,$00] 'see page 46 in ISD1700 Design Guide
HIGH SS

LOW LED          'LED OFF

END

```

### Example 3 – Playing based on START and STOP Memory Locations (Using SHIFTOUT Command)

```
' Name:   PLAYING the ISD17150-SO.BAS
' Author: Ian Burn
' Date :  13/02/2009
' Version: 3
' Notes: 149 Words   Special thanks to Charles Leo of Micro Engineering Labs
'        MODE 4 in SHIFTOUT command sets clock high and Least Sig Bits sent first
'
'References:
'        ISD1700 Design Guide.PDF - Timing diagrams KEY TO SUCCESS
'        ISD1700 Data Sheet.PDF
'        ISD1700 Addresses AN-CC1005.PDF - Essential for finding address locations
'
'-----
'OVERVIEW
'-----
' Objective is to PLAY a message, starting at memory location 16 and ending at memory location 81.
' Two commands are sent: POWER-UP and SET_PLAY
' All timing and data transmission are done via the SHIFTOUT command in PBP
'
'-----
'CONFIGURATION
'-----
TRISIO=%001000   'Ports 0,1,2,4 and 5 as OUTPUTS
CMCON0=7        'COMPARATOR OFF
ANSEL=0         'all ports set as digital

@ DEVICE PIC12F683, WDT_ON, MCLR_ON, BOD_OFF, PWRT_ON, PROTECT_OFF

'-----
'DEFINE VARIABLES
'-----

'PORTS

SS VAR GPIO.0   'Pin #7 output
SCLK VAR GPIO.1 'Pin #6 output
MOSI VAR GPIO.2 'Pin #5 output
'GPIO.3 MCLR    'Pin #4 input
LED VAR GPIO.4  'Pin #3 output
'GPIO.5 not used 'Pin #2 input

'Other Variables

x VAR word      'misc. counter

'-----
'PROGRAM
'-----

START:
HIGH LED        'LED ON during the program

HIGH SS
LOW MOSI
HIGH SCLK
```

```

'POWER_UP
LOW SS
SHIFTOUT MOSI, SCLK, 4, [$01,$00] 'see page 41 in ISD1700 Design Guide
HIGH SS

PAUSE 10 'Depending on the command that follows this isn't always needed
        'but is likely good insurance

'SET_PLAY
'start at memory 16 and end at memory 81 Bit 4 Byte 1 for LED on Demo Board
LOW SS
SHIFTOUT MOSI, SCLK, 4, [$90,$00,$10,$00,$51,$00,$00] '$90 = SET_PLAY w LED, 16= $10 and
        ' 81 = $51

HIGH SS
pause 8000 'allow 8 seconds to play

LOW LED 'LED OFF

END

```

For all other commands refer to the ISD1700 Design Guide. This note is only intended to assist in the initial communications between the PIC 12F683 and the ISD17150 ChipCorder Demo board.

## Acknowledgements

Special thanks to:

Jerry Wang of Nuvoton who helped with the 'bit banging' approach that led to the discovery of a faulty Demo Board.

and

Charles Leo of Micro Engineering Labs who convinced me that SHIFTOUT would indeed work with the ISD1700 series Demo Board as long as it isn't faulty!

and

DigiKey for replacing the faulty board with a one-day turn around.

## References

ISD1700 Design Guide available on request from Nuvoton or see:

[http://www.techs-store.com/support/manual/isd1700/isd1700\\_eman.pdf](http://www.techs-store.com/support/manual/isd1700/isd1700_eman.pdf)

ISD1700 Addresses AN-CC1005.pdf See:

[http://www.nuvoton-usa.com/products/isd\\_products/chipcorder/applicationbriefs/AN-CC1005\\_a.pdf](http://www.nuvoton-usa.com/products/isd_products/chipcorder/applicationbriefs/AN-CC1005_a.pdf)

ISD-COB 1700 Demonstration Board See:

<http://www.nuvoton.com/hq/enu/ProductAndSales/ProductLines/ConsumerElectronicsIC/ISDVoiceIC/ISDChipCorder/Demo+Boards.htm#ISDCOB17XXX>

Demo Board Source: DigiKey part number: ISD-COB17150-ND (\$17.65 CAN)